
Except I'm not actually going to be using The Sims for testing.
It's just an analogy
Sorry if the talk just got 98% less awesome for some.



The Sims is this game about building a home and life for these little simulated people, right?

Well, maybe for a while.  But eventually, power corrupts.



You create some horrifying near-human analogue



Try your hand at decor...



Art...
Right around when you realize you can't make porn in The Sims
you get bored



And then begins the torturing.



Until they beg for death



And inevitably curl up and die.



And then you start the process all over again!



And that's what you do with test data



Create it



Mutate it



Torture it



Delete it



Testing situation:
Your system needs a bunch of data before it can do anything interesting



A lot of commercial web sites have this problem
Say, IMDB.
I happen to know they use Perl, and I like movies.
Before you can test an entry for a film you need all sort of dependent data...
Generating all that for each test is cumbersome
Some people instead copy the production database.
But this is brittle
Data might not be interesting enough
It might get dirty in the middle of a run and effect later tests
Change the test data for one test and you might break another.
God forbid the schema changes.
"The Sims" technique is another option



The Sims Technique

There's not much to it.
This is a technique I came up with
while working with Grant Street Group





1.  What data do you need?

Just for this test.
What makes up a "movie"?





Film
    Title

    Directors

    Writers

    Release Year

    Genres
    
    Tagline

    Plot

    User Rating

    Photos

An Overview object would contain all this
And those objects would contain...





Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
    
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image

...all this.
Dependent tree of data.
That's a lot of objects to initialize before you can get a test going.
You're not going to want to write tests.



2.  Find the bottom

Or a "leaf"
Find some data that doesn't depend on any more data.



Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
        Text
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image

Well, there they are.



3.  Write a function to 
generate random, valid data



Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
    
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image

Let's start with "text"
We can generate random text.



my @Texts = (
  "Basset hounds got long ears",
  "Oodles of poodles jump der strudel",
  "I woke up naked, handcuffed to the bus stop, again",
);

sub rand_text {
    return $Texts[rand @Texts];
}

Well that was easy.
Doesn't matter that it's nonsense, as long as it's valid.
In fact, it's good that it's nonsense because it's unexpected.
Exercise the code better
You can always add more later.



Random, Interesting
Data

You think about making your data interesting once
Then it gets used all over the place
And it'll find its way into all sorts of interesting nooks
Ferret out all sorts of interesting bugs



4.  Generate an object with 
the random data



Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
    
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image

Well, there they are.



Person
    Name:   Text
    DoB:    Datetime
    Trivia: [Text, ...]



use DateTime;

sub rand_datetime {
    my $date = DateTime->new(
        year  => rand_year(),
    );

    $date->add( days    => int rand 366 );
    $date->add( seconds => int rand 60 * 60 * 24 );

  return $date;
}

DateTime made this so much easier.
And it can handle our range of years.



sub sim_person {
    return Person->create(
        name   => rand_text(),
        dob    => rand_datetime(),
        trivia => [map rand_text(), 0..rand 19],
    );
}

And now we can test it!



package My::Sims;

use base qw(Exporter);
our @EXPORT = qw(
    sim_person
);

our @EXPORT_OK = qw(
    rand_datetime
    rand_text
    rand_year
);

our %EXPORT_TAGS = (
    rand  => [grep /^rand_/, @EXPORT_OK, @EXPORT],
    sims  => [grep /^sim_/,  @EXPORT_OK, @EXPORT],
);

sub sim_person {...

Shove it in a module



use Test::More tests => 2;
use My::Sims;

my $person = sim_person();
isa_ok $person, "Person";

is $person->name, "???";

Uh oh
How do you test random data?
We need a way to nail parts of it down.



sub sim_person {
    return Person->create(
        name   => rand_text(),
        dob    => rand_datetime(),
        trivia => [map rand_text(), 0..rand 19],
    );
}



sub sim_person {
   my %defaults = (
        name   => rand_text(),
        dob    => rand_datetime(),
        trivia => [map rand_text(), 0..rand 19],
    );

    return Person->create(
        %defaults, @_
    );
}



5.  Write some tests with it!



use Test::More tests => 2;
use My::Sims;

my $person = sim_person();
isa_ok $person, "Person";

is $person->name, "???";



use Test::More tests => 2;
use My::Sims;

my $person = sim_person(
    name => "Weng Weng"
);
isa_ok $person, "Person";

is $person->name, "Weng Weng";

We've nailed down the name.
The rest is left as random or defaults.
We don't care about them.



Random data exercises 
code very well



6.  Combine to make 
complex data



Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
        Text
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image



sub sim_film {
    my %defaults = (
        title        => rand_text(),
        directors    => [map sim_person(), 0..rand 3],
        writers      => [map sim_person(), 0..rand 6],
        release_year => rand_year(),
        genres       => [map rand_text(),  0..rand 12],
        tagline      => rand_text(),
        plot         => rand_text(),
        user_rating  => rand 10,
        photos       => [map rand_image(), 0..rand 20],
    );

    return Film->create(
        %defaults, @_
    );
}



use Test::More 'no_plan';
use My::Sims;

my $film = sim_film(
    title => "For Your Height Only",

    plot  => <<'END_PLOT',
A secret agent midget runs around
punching people in the wing-wang.
END_PLOT
);

...

And now I have a populated, "interesting" film for testing.
Only the parts I care about for this test are fixed
The rest are random but valid
Surprise testing



use Test::More 'no_plan';
use My::Sims;

my $film = sim_film(
    title     => "For Your Height Only",

    directors => [sim_director(
        name => 'Eddie Nicart'
    )],
);

...

You can combine them.



What do you do when it 
fails?

The data is random.
So you get heisen-failures.
You need some way to repeat the same failure.



our $Seed = $ENV{TEST_SIMS_SEED} || (time ^ $$);
srand $Seed;

END { print "# TEST_SIMS_SEED=$Seed\n" }

Doesn't matter that it's not a very secure seed.
Just that it's repeatable.



$ prove -lv t/person.t
1..4
ok 1 - isa Person
ok 2 - title
not ok 3 - user rating
ok 4 - ham
# TEST_SIMS_SEED=8273872



$ TEST_SIMS_SEED=8273872 prove -lv t/person.t
1..4
ok 1 - isa Person
ok 2 - title
not ok 3 - user rating
ok 4 - ham
# TEST_SIMS_SEED=8273872



7.  Improve the data



Make it more interesting

You get a bug reported and it's because of some unexpected data...



my @Texts = (
  "Basset hounds got long ears",
  "Oodles of poodles jump der strudel",
  "I woke up naked, handcuffed to the bus stop, again",
);

sub rand_text {
    return $Texts[rand @Texts];
}



my @Texts = (
  "Basset hounds got long ears",
  "Oodles of poodles jump der strudel",
  "I woke up naked, handcuffed to the bus stop, again",
  "1942",
  "@#$!",
  "괴물",
);

sub rand_text {
    return $Texts[rand @Texts];
}



Specialize the data



my @Genres = (
    "Action",
    "Comedy",
    "Midget",
    "James Bond Spoof",
);

sub rand_genre {
    return $Genres[rand @Genres];
}

And if you need to later, you can specialize.



Keeps you from being 
overwhelmed



Easy & fast to write tests

People don't want to write tests
So you have to make them as fast and easy as possible
It also leave more time to write better, more through tests
And not scaffolding



Repaying Tech Debt

If you start doing this at the beginning of the project, it'll build up as your code does.
Otherwise, you might have a lot of work to do.
You're paying back technical debt.
This is why you start at the bottom and work up.
To keep yourself sane.



Test::Sims?

Look on CPAN for Test::Sims... soon.
Take care of random number fixing
Maybe eliminate rand_* and exporter scaffolding.



Review



1.  What data do you need?

Just for this test.
What makes up a "movie"?



Film
    Title

    Directors

    Writers

    Release Year

    Genres
    
    Tagline

    Plot

    User Rating

    Photos

An Overview object would contain all this
And those objects would contain...



2.  Find the bottom

Or a "leaf"
Find some data that doesn't depend on any more data.



Film
    Title
        Text
    Directors
        Person
    Writers
        Person
    Release Year
        Year
    Genres
        Text
    Tagline
        Text
    Plot
        Text
    User Rating
        Decimal
    Photos
        Image

Well, there they are.



3.  Write a function to 
generate random, valid data



my @Texts = (
  "Basset hounds got long ears",
  "Oodles of poodles jump der strudel",
  "I woke up naked, handcuffed to the bus stop, again",
);

sub rand_text {
    return $Texts[rand @Texts];
}

Well that was easy.
Doesn't matter that it's nonsense, as long as it's valid.
In fact, it's good that it's nonsense because it's unexpected.
Exercise the code better
You can always add more later.



4.  Generate an object with 
the random data



sub sim_person {
   my %defaults = (
        name   => rand_text(),
        dob    => rand_datetime(),
        trivia => [map rand_text(), 0..rand 19],
    );

    return Person->create(
        %defaults, @_
    );
}



5.  Write some tests with it!



use Test::More tests => 2;
use My::Sims;

my $person = sim_person(
    name => "Weng Weng"
);
isa_ok $person, "Person";

is $person->name, "Weng Weng";

We've nailed down the name.
The rest is left as random or defaults.
We don't care about them.



6.  Combine to make 
complex data



sub sim_film {
    my %defaults = (
        title        => rand_text(),
        directors    => [map sim_person(), 0..rand 3],
        writers      => [map sim_person(), 0..rand 6],
        release_year => rand_year(),
        genres       => [map rand_text(),  0..rand 12],
        tagline      => rand_text(),
        plot         => rand_text(),
        user_rating  => rand 10,
        photos       => [map rand_image(), 0..rand 20],
    );

    return Film->create(
        %defaults, @_
    );
}



7.  Make the data more 
interesting



my @Texts = (
  "Basset hounds got long ears",
  "Oodles of poodles jump der strudel",
  "I woke up naked, handcuffed to the bus stop, again",
  "1942",
  "@#$!",
  "괴물",
);

sub rand_text {
    return $Texts[rand @Texts];
}

It's easy to expand the data to be more interesting.



Thanks





Questions?


